
Goal of the lecture

Illustrate the use of concurrent communicating ASMs to construct
a ground model for a real-life commercial business process

namely the business logic of the Sardex mutual credit system.

The SARDEX S.p.A. company operates an electronic, B2B, zero-interest
mutual credit system on the island of Sardinia since 2009.

Innovative system feature: distribute to the circuit members the power
to create zero-interest credit money (1 Sardex credit = 1 Euro),
providing an alternative to credit money creation through bank loans.

The development of a new decentralized transactional and ledger
system architecture led to rigorously specifying also the underlying
business logic, prior to its implementation, using communicating ASMs.

Part of the work was done in the European Commission Project
INTERLACE https://www.interlaceproject.eu/ (2017-2018).

Why INTERLACE used ASMs to specify the architecture

The ASM method allows one to achieve the following goals:

to provide an accurate requirements capture

– building a common understanding and a guide for the implementation

to provide a rigorous practical refinement process down to code

– so that the coding can be split efficiently into effectively controllable
separate steps rather than achieving the full redesign in one step

to provide an efficient change management

– to handle the evolution of requirements, due to continuously
emerging new or improved system functionalities

such that at any development level the model components are analysable
both mathematically and experimentally—since they are precise and
executable—and are easily extendable (via refinements)

i.e. such that at each refinement stage the current implementation can
be verified and validated against the more abstract version it refines,
thus providing a reliable link to the original requirements.

Copyright Egon Börger, released under Creative Commons BY-NC-SA International Licence 4.0 3



Selection of characteristic Sardex service operations

Sardex members—companies, their employees, individual consumers, but
also members of associated networks—interact with the Sardex network
server by requesting services concerning Sardex accounts owned by
members. Here we illustrate 4 core services (a bare minimum):

handling credit or debit requests, the two central Business-to-Business
(B2B) operations between companies to transfer the amount for a
purchase from the buyer’s Sardex account—which can go negative up
to some limit with a 0% interest rate—to the seller’s Sardex account

handling Business-to-Consumer (B2C) operations where a customer
purchase at a retail is paid in Euro (EUR) or Sardex (SRD) with
Sardex acting as intermediary (honored by a retailer’s prepaid fee)

Alerts about low/high balance, high sale volume, low prepayment,
etc.

debt record tracking , a refinement of credit/debit ops

NB. Selection covers a tiny but characteristic subset of all Sardex opns.

Focus of the illustrative HandleSardexOpns server model

functional requirements level of abstraction to express the system’s
business logic in user (not programming lg or implementation) terms

proceeding by stepwise refinements of abstractions wherever possible

– to efficiently support integration of new (or change of) requirements

• here illustrated by adding debt record tracking to credit/debit opns

– to smoothly combine control and data features of the system where
behaviorally relevant (a crucial requirement for real-life BPs!)

component-based description of Sardex services

– defining for each service one component to HandleSardexOpns

– further splitting each service component into subcomponents

separating the spec of the scheduler from that of component behavior

– using parallelism, limited by dependency constraints—imposing
sequential execution—only where required by the business logic

to leave the space open for efficient implementations of the model

Roadmap to construct the Sardex model

Sardex server component HandleSardexOpns

– parallel component structure of the model

– actors and data involved in the services we model here:

•member groups and their interaction constraints (group transfer
type constraints)

• account sets and their connectivity constraints

– behavior of HandleSardexOpns (their semantical meaning)

• Components of Credit op rule HandleCreditTransfer

• Components of Debit op rule HandleDebitTransfer

• Components B2C op rule HandleB2cTransfer

• Components of Alerts

MemberOpns component to Send/Receive input to/from the server

Refinement example: integrate debt record tracking

–CreateDebtEntry and UpdateDebtRecord

HandleSardexOpns structure for the selected service opns

HandleSardexOpns = -- parallel composition of components

HandleCreditTransfer HandleDebitTransfer

HandleB2cTransfer Alerts

—————————————————————————————–

HandleCreditTransfer = -- 2-phase request/response pattern

HandleCreditPreviewReq HandleCreditPerformReq

HandleDebitTransfer = -- 3-phase request/response pattern

HandleDebitPreviewReq HandleDebitPerformReq

HandleDebitAckCompletion

HandleB2cTransfer =

HandleB2cEur HandleB2cSrd RechargePrepaid

HandleAlerts =

HighBalanceAlert,LowPrepayAlert, . . .

Some typical actors which trigger HandleSardexOpns

To HandleSardexOpns is triggered by actors of user groups:

Company -- actors which participate only in B2B opns

Retail -- actors which participate only in B2C operations

Full -- actors with both Company and Retail functionality

Consumer -- individuals which participate only in B2CEUR opns

Consumer Verified -- registered consumers with additional B2CSRD opn

Mngr = {mngr} -- mngr acts for Sardex as a company

Group =

{Company ,Retail ,Full ,Consumer ,Consumer Verified ,Mngr}
We skip for example Employees, not-yet-cleared (called Welcome
members), suspended (called On Hold) users or users of other circuits.

Groups come with group profile metadata (their ‘state’) and transfer
type constraints on the groups among which transfers are allowed.

Three characteristic accounts HandleSardexOpns manages

Each user has a set Account(user ) of Sardex accounts each of which
the user is the owner (acc) of, at most one account per account type.

Here we illustrate 3 characteristic account types (sets of accounts):

AcctType = {CC , Income,Prepaid}

CC : contains SRD-credit accounts cc(user ) used to pay for purchases
in SRD currency (for each user ∈ group of any group ∈ Group)

The sets Income resp. Prepaid of Euro-accounts contain for each
user ∈ Retail ∪ Full additionally two accounts:

– income(user ) recording B2CEUR payments made by consumers

– prepaid(user ) recording the Euro fee due by the user to the
intermediary—the Sardex company—for each B2CEUR operation

Accounts come with account metadata (functions) and connectivity
constraints on types of accounts between which an operation is allowed.

Copyright Egon Börger, released under Creative Commons BY-NC-SA International Licence 4.0 9



Typical CC -account metadata

Sardex CC -accounts can be used by members to

pay another member in SRD, up to a creditLimit , for a purchase

be payed by another member in SRD, up to an upperBalanceLimit ,
for goods sold to that other member

These limits protect the circuit and its members:

in case of no upperBalanceLimit limit an extremly positive balance
could make it difficult for the member to spend the credits by
purchases, inviting not to sell any more (risk to become inactive in the
circuit)

in case of no creditLimit an extremly negative balance could make it
difficult for the member to proceed with further purchases (to avoid a
too high debt) with the risk to become inactive in the circuit if no
credits arrive by sales.

CC -account functions/predicates

balance : Real -- may become negative

NB. This function has an exclusive access constraint (see below).

creditLimit : Nat -- limiting how far account can go negative, to buy

availableBalance = balance + creditLimit ∈ Nat

LowBalanceAlert iff availableBalance < lowBalanceAlert

-- small amount of money left for further purchases in SRD

upperBalanceLimit ∈ Nat \ {0}
-- limiting how far account can go positive, to sell

HighBalanceAlert iff balance > 0 and

upperBalanceLimit − balance < highBalanceAlert

-- small space left for further sales in SRD

Other similar CC -account metadata

There are other already existing or planned CC -account functions, e.g.

– saleCapacity indicating the maximum yearly SRD volume the
member committed to be willing to sell with payment in SRD

• used to calculate the annual Sardex fee

– saleVolume denoting the current total volume of sales already made
and paid in SRD (per year, we notationally suppress the param)

• NB. This function has an exclusive access constraint (see below).

– availableSaleCapacity = saleCapacity − saleVolume

• usable to issue a HighVolumeAlert if availableSaleCapacity goes
below a highVolumeAlert value

· possibly resulting in an update of saleCapacity and annual fee

We skip various restrictions on account functions, e.g. that

– Prepaid accounts are without credit (i.e. creditLimit = 0)

– Domu accounts have no saleCapacity

– brokers may update certain functions, e.g. setting creditLimit := 0

Typical group profile metadata (ASM functions/predicates)

creditPercent denotes the percentage of SRD-payments accepted by a
Full or Retail member for B2B-transactions of value>1000 Euro

– NB. In a previous system version this holds also for B2E-transactions

acceptanceRate defines the percent rate of the total purchase value at
which Full or Retail members accept SRD currency from a registered
consumer (i.e. for a B2C-transaction)

rewardRate of Full and Retail members indicates the percentage of
reward the member offers in SRD for a B2CEUR purchase

fee locations for Company , Retail and Full members

– for Retail a B2CEuroFee fctn indicating the fee for B2CEUR sales

• the percentage the retailer pays to Sardex for each B2CEUR sale

– for Company an InterTradeEuroFee fctn indicating the fee for
inter-circuit sales in the non-Euro circuit currency (SRD, VTX, etc).

– Full members have 2 functions InterTradeEuroFee, B2CEuroFee.

Group transfer type constraints on Credit/Debit transactions

The constraints depend on the operation, the currency and the group:

TT : {Credit ,Debit}× {SRD ,EUR}×Group → {G | G ⊆ Group}
We split the transfer type function TT as follows into subfunctions

TTop,cur : Group → {G | G ⊆ Group}
For each operation, each cur rency and each fromGroup of buyers:

TTop,cur (fromGroup) defines the set toGroups of groups of sellers
which are admitted for the transfer operation in the cur rency

– namely of those groups whose members are allowed to receive a
transfer, concerning the operation in the cur rency, to one of their
accounts from an account of a fromGroup member

• under separately defined appropriate account connectivity
constraints and further constraints on the amount of the transfer

NB. Separating TT -subfunctions and constraints on groups, accounts
and transfer amounts supports modularity and simplicity concerns.

Group Transfer Type Constraints for TTCredit ,SRD

A Credit SRD -op can be started only by members of

– Company or Full or Mngr or Consumer Verified

MayTriggerCreditOp(mbr ) iff
group(mbr ) ∈ {Company ,Full ,Mngr ,Consumer Verified}

The target groups allowed for a Credit SRD -op are defined by:

– every Company ∪ Full ∪Mngr member can trigger a Credit
SRD -op to a member of Company ∪ Full ∪Mngr

– Consumer Verified members can trigger a Credit SRD -op to
Retail ∪ Full -members (NB. We simplify these opns below)

Equationally this is expressed as follows:

TTCredit ,SRD(G) = {Company ,Full ,Mngr} if G ∈ {Company ,Full ,Mngr}

{Retail ,Full} if G = Consumer Verified

Group Transfer Type Constraints for TTDebit ,SRD

For Debit SRD -ops the following source/target groups are allowed:

– Every Company ∪ Full ∪Mngr member MayTriggerDebitOp (in
SRD) that draws on a member of Company ∪ Full .

– Some other constraints (there are more) we do not use in this model:

• Every member of Retail ∪ Full can start a Debit SRD -op that
draws on any member of Consumer Verified

•mngr can start a Debit SRD -op that draws on a Retail member

TTDebit ,SRD(G) = -- we use only the first clause
{Company ,Full ,Mngr} if G ∈ {Company ,Full}

{Retail ,Full} if G = Consumer Verified

Mngr if G = Retail

Account connectivity constraints

Since for reasons of simplicity we model B2C-ops not as Credit/Debit
ops, it suffices here to consider user-initiated CC -transactions:

An SRD-Credit transfer op is triggered by a debitor, starts at a
fromAcc ∈ CC and has as allowed toAcc a CC account.

An SRD-Debit op is triggered by a creditor, starts from a debitor’s
fromAcc ∈ CC and goes to the creditor’s toAcc ∈ CC .

AccTop,cur : AccountType → {Acct | Acct ⊆ AccountType}
AccTCredit ,SRD(CC ) = AccTDebit ,SRD(CC ) = {CC}

MayStartCreditOpns(acct) iff acct ∈ CC

MayStartDebitOpns(acct) iff acct ∈ CC

NB. Such definitions (of ASM functions, predicates, rules, etc.) by case
distinction support componentwise extensions to new requirements

– e.g. to new account type arguments for AccTCredit ,SRD we do not
use here, which are similar to the above clauses for TTDebit ,SRD

Request/Response pattern of Credit/Debit operations

Credit resp. Debit are push resp. pull ops, triggered by a debitor resp.
creditor, to transfer amounts from a debitor’s to a creditor’s account.

Credit/Debit ops follow a request/response pattern between the actors:

– members equipped with the MemberOpns program

– the Sardex server which executes the HandleSardexOpns pgm

We abstract from communication Channels, namely a website, a
mobile Phone or a standard point-of-sale (POS) terminal retailers use
for EUR transactions or to route SRD transactions via an API.

– Reason: The business logic effect of all Credit/Debit op elements we
define below is independent of the particular value of their Channel
parameter and of the particular communication mechanism used.

Credit/Debit ops follow a multiple-phase request/response pattern:

– Preview, where only group/account-related access rights are checked

– Perform, where the critical transfer amount constraints are checked

HandleCreditTransfer request/response control flow

debitor sardex debitor

CreditPreviewReq → CreditTypeCheck

PermitPerformReq ← if ok if fail → FailureMsg

CreditPerformReq → CreditTypeCheck

if ok ↓ if fail → FailureMsg

CreditAmount
LimitsCheck

if ok ↓ if fail → FailureMsg

ConfirmationMsg ← CompleteTransaction

NB. Arrows indicate the sequential control and data flow.

2-phase structure of the HandleCreditTransfer service

The Sardex server, upon receiving a member’s (a debitor’s)

CreditPreviewRequest : will HandleCreditPreviewReq

CreditPerformRequest : will HandleCreditPerformReq

HandleCreditTransfer =

HandleCreditPreviewReq

HandleCreditPerformReq

The server can execute at any moment any of these two rules whose
execution depends only on the parameters with which they are called.

– But the rules are defined in such a way that for a specific user request
parameter, HandleCreditPerformReq can be triggered only
if HandleCreditPreviewReq yields a positive check result.

For the transactional part of HandleSardexOpns concerning
accounts it is assumed for the model that access to the involved
accounts is exclusive (see below), as usual for db-transactions.

How HandleCreditPreviewReq interacts with members

HandleCreditPreviewReq triggers a CreditTypeCheck
concerning the group/account related access rights wrt the parameters
of the member’s CreditPreviewRequest

– upon a positive check result the server will PermitPerformReq

– otherwise it will send a constraint violation msg to the member

HandleCreditPreviewReq((from, to, amount),mbr ) =

let t = (credit , from, to, amount) -- the transfer params

if Received(CreditPreviewReq(t), from mbr ) then

CreditTypeCheck(t ,mbr ,PermitPerformReq(t))

Consume(CreditPreviewReq(t))

where

PermitPerformReq(t) = -- NB. mbr = owner (from)

Send(YouMayTriggerPerformReq(t), to mbr )

How HandleCreditPerformReq interacts with members

HandleCreditPerformReq repeats CreditTypeCheck

upon a positive outcome it will TryToCompleteCreditOpn by
a CreditAmountLimitsCheck

– to CompleteTransaction definitely, in the positive case

HandleCreditPerformReq((from, to, amount),mbr ) =

let t = (credit , from, to, amount) -- the transfer params

if Received(CreditPerformReq(t), from mbr ) then

CreditTypeCheck(t ,mbr ,TryToCompleteCreditOpn(t))

Consume(CreditPerformReq(t))

where TryToCompleteCreditOpn(t) =

CreditAmountLimitsCheck(t ,CompleteTransaction(t))

CreditTypeCheck((from, to, amount),mbr ,Proceed)

let t = (credit , from, to, amount) -- the transfer params

if mbr = owner (from) and MayTriggerCreditOp(mbr ) then

-- check group transfer type constraints

if forsome g ∈ TTCredit ,SRD(group(mbr )) owner (to) ∈ g then

if MayStartCreditOpns(from) then

-- check account connectivity constraints

if acctSet(to) ∈ AccTCredit ,SRD(acctSet(from)) then

Proceed(t) -- in positive case PermitPerformReq(t)

else Send(ErrMsg(CreditTargetAcctError (t)), to mbr )

else Send(ErrMsg(CreditSourceAcctError (t)), to mbr )

else Send(ErrMsg(CreditTargetGroupError (t)), to mbr )

else Send(NotAcctOwnerOrCreditSourceGroupError (t), to mbr )

CreditAmountLimitsCheck((from, to, amount),Proceed)

let t = (credit , from, to, amount) -- the transfer params

if CanBeSpentBy(from, amount) then

if CanBeCashedBy(to, amount) then

if HasSellCapacityFor (amount , to) then Proceed(t)

-- namely to CompleteTransaction

else Send(ErrMsg(CapacityViolation(t)), to owner (from))

else Send(ErrMsg(UpperBalanceLimitErr (t)), to owner (from))

else Send(ErrMsg(AvailBalanceViolation(t)), to owner (from))

where

CanBeSpentBy(from, a) iff availableBalance(from) ≥ a

CanBeCashedBy(to, a) iff

balance(to) + a ≤ upperBalanceLimit(to)

HasSellCapacityFor (a, to) iff a ≤ availableSaleCapacity(to)

CompleteTransaction component

The abstract fctn transact encodes the transfer info that is appended
to the Ledger and results in an update of the critical balance and
saleVolume locations (constrained, as usual for db transactions, by an
exclusive access condition in TryToCompleteCreditOpn).

CompleteTransaction(transfer ) =

let (credit , from, to, amount) = transfer

Append(transact(transfer , now ),Ledger ) -- timestamp now

Send(Confirmed(transfer ), to owner (from))

where Append(transact(transfer , now ),Ledger ) includes

balance(from) := balance(from)− amount

balance(to) := balance(to) + amount

saleVolume(to) := saleVolume(to) + amount

NB. Here we abstract from the underlying concrete db-scheme.

3-phase structure of the HandleDebitTransfer service

The Sardex server—note the symmetry of Credit/Debit ops—

upon receiving a member’s (a creditor’s)

– DebitPreviewRequest : will HandleDebitPreviewReq

– DebitPerformRequest : will HandleDebitPerformReq

upon a positive DebitAmountLimitsCheck will interact with the
debitor to RequestDebitAcknowledgement (except for small
amounts) before it can HandleDebitAckCompletion

HandleDebitTransfer =


HandleDebitPreviewReq

HandleDebitPerformReq

HandleDebitAckCompletion

HandleDebitAckCompletion splits into two suboperations:

a DebitAckCompletion if the debitor’s ack arrives in time

a DebitAckRequestExpired component, otherwise

HandleDebitTransfer request/response flow: phase 1-2

creditor sardex debitor

DebitPreviewReq → DebitTypeCheck

PermitPerformReq ← if ok else FailureMsg

DebitPerformReq → DebitTypeCheck

if ok ↓ else FailureMsg

DebitAmountLimitsCheck

if ok ↓ else FailureMsg

RequestDebitAck

if LargeAmount then → AckReq

ConfirmationMsg ← else CompleteTransaction

HandleDebitTransfer request/response flow: phase 3

creditor sardex debitor

RejectMsg ← if ExpiredAckReq → ExpireMsg

DebitAckCompletion ← AnswerMsg

RejectMsg ← if RejectAnsw else↓

FinalDebitAmtLimitsCheck

RejectMsg ← if CheckFails → FailureMsg

else ↓

Confirmation ← CompleteTransaction

How HandleDebitPreviewReq interacts with members

HandleDebitPreviewReq triggers a DebitTypeCheck
concerning the group/account related access rights which correspond
to the parameters of the creditor’s DebitPreviewRequest

– upon a positive check result the server will PermitPerformReq

– otherwise it will send a constraint violation msg to the creditor

HandleDebitPreviewReq((from, to, amount), creditor ) =

let t = (debit , from, to, amount) -- the transfer params

if Received(DebitPreviewReq(t), from creditor ) then

DebitTypeCheck(t , creditor ,PermitPerformReq(t))

Consume(DebitPreviewReq(t))

where

PermitPerformReq(t) = -- NB. creditor = owner (to)

Send(YouMayTriggerPerformReq(t), to creditor )

DebitTypeCheck((from, to, amount),mbr ,Proceed)

let t = (debit , from, to, amount) let debitor = owner (from)

if mbr = owner (to) and MayTriggerDebitOp(creditor ) then

-- check group transfer type constraints for the creditor mbr

if forsome g ∈ TTDebit ,SRD(group(debitor )) mbr ∈ g then

if MayStartDebitOpns(to) then

-- check account connectivity constraints

if acctSet(from) ∈ AccTDebit ,SRD(acctSet(to)) then

Proceed(t) -- in positive case PermitPerformReq(t)

else Send(ErrMsg(DebitTargetAcctError (t)), to mbr )

else Send(ErrMsg(DebitSourceAcctError (t)), to mbr )

else Send(ErrMsg(DebitTargetGroupError (t)), to mbr )

else Send(NotAcctOwnerToReceiveDebit(t), to mbr )

How HandleDebitPerformReq interacts with members

HandleDebitPerformReq repeats DebitTypeCheck

upon a positive outcome it will TryToCompleteDebitOpn by a
DebitAmountLimitsCheck

– to RequestDebitAck, from the debitor, upon a positive check
outcome and then HandleDebitAckCompletion

HandleDebitPerformReq((from, to, amount),mbr ) =

let t = (debit , from, to, amount) -- the transfer params

if Received(DebitPerformReq(t), from creditor ) then

DebitTypeCheck(t , creditor ,TryToCompleteDebitOpn(t))

Consume(DebitPerformReq(t))

where TryToCompleteDebitOpn(t) =

DebitAmountLimitsCheck(t ,RequestDebitAck(t))

DebitAmountLimitsCheck component

DebitAmountLimitsCheck((from, to, amount),Proceed) =

let t = (debit , from, to, amount)

let debitor = owner (from)

let creditor = owner (to)

if CanBeSpentBy(from, amount) then

if CanBeCashedBy(to, amount) then

if HasSellCapacityFor (amount , to) then Proceed(t)

-- namely to RequestDebitAck

else Send(ErrMsg(SellCapacityViolation(t)), to creditor )

else Send(ErrMsg(UpperBalanceLimitErr (t)), to creditor )

else Send(ErrMsg(AvailBalanceViolation(t)), to debitor )

Send(ErrMsg(DebitorHasSomeProblemWith(t)), to creditor )

RequestDebitAck behavior

RequestDebitAck does CompleteTransaction without
further ado if the amount is Small (less than 100).

– The definition of CompleteTransaction is extended to Debit
transfers (the confirmation msg addressee changes):

CompleteTransaction(debit , from, to, amount) =
let transfer = (debit , from, to, amount) -- symmetric to credit
Append(transact(transfer , now ),Ledger )
Send(Confirmed(transfer ), to owner (to))
saleVolume(to) := saleVolume(to) + amount

DebitAmountLimitsCheck(t), if Small(amount) including
CompleteTransaction(t), requires exclusive db-access.

For other amounts it creates a OneTimePassword for an agreement
request sent to the debitor and records the transaction as pending .

HandleDebitAckCompletion can be executed when an answer
msg for the pendingT ransaction arrives or when Expired(otp).

RequestDebitAck component

RequestDebitAck(transfer ) =

let (debit , from, to, amount) = transfer

let debitor = owner (from) creditor = owner (to)

if Small(amount)

then CompleteTransaction(transfer )

else

let otp =new (OTP)

let pendgT = (otp, (transfer ))

birthTime(otp) := now

Insert(pendgT ,PendingTransact)

status(pendgT ) := pending

Send(AgreementReq(pendgT ), to debitor )

Debit transaction phase 3: HandleDebitAckCompletion

HandleDebitAckCompletion splits into two suboperations:

a DebitAckCompletion if the debitor’s ack arrives in time

a DebitAckRequestExpired otherwise

For DebitAckCompletion there are 3 disjoint cases to consider
when an AnswerMsg(pendingT ) arrives from the debitor:

1. no such pendingT exists (maybe any more) in PendingTransact or
its otp is Expired . Such a msg is simply discarded.

2. AnswerMsg(pendingT ) is a DebitRejectMsg . Then pendingT is
moved from PendingTransact to RejectedTransact and its status
updated to rejected .

3. AnswerMsg(pendingT ) is a DebitAcceptMsg . Then, after a
FinalDebitAmtLimitsCheck with positive outcome, the
transaction status can be updated from pending to performed to
CompleteTransaction.

DebitAckCompletion component

DebitAckCompletion =

if Received(AnswerMsg(pendgT ), from debitor ) then

if pendgT ∈ PendingTransact and not Expired(pendgT ) then

let (otp, (debit , from, to, a)) = pendgT creditor = owner (to)

if AnswerMsg(pendgT ) = DebitRejectMsg(pendgT )

then

RecordRejected(pendgT )

Send(RejectionInfo(Debit , a, debitor ), to creditor )

else RecordAccepted(pendgT )

FinalDebitAmtLimitsCheck

(pendgT ,CompleteTransaction(pendgT ))

Consume(AnswerMsg(pendgT ))

DebitAckCompletion macros

Expired(pendingT ) iff

now − birthtime(pendingT ) > OtpLifetime

RecordRejected(t) = -- analogous for Accepted

status(t) := rejected

Delete(t ,PendingTransact) Insert(t ,RejectedTransact)

CompleteTransaction(pendingT ) =

let (otp, transfer ) = pendingT

CompleteTransaction(transfer )

status(pendingT ) := performed

NB. The two versions of CompleteTransaction for pending
transactions and for transfers differ by the number of parameters.

FinalDebitAmtLimitsCheck

FinalDebitAmtLimitsCheck is a refinement of the above defined
component DebitAmountLimitsCheck:

into the failure cases an additional clause is inserted which makes the
pending Debit transaction rejected and informs the creditor about the
reason for rejection.

FinalDebitAmtLimitsCheck, in case of a positive check result
including CompleteTransaction, is required to be transactional

i.e. with exclusive access to the two critical functions balance and
salesVolume.

– For some examples of Mutex algorithms one could use to satisfy the
exclusive access requirement see the lecture notes
http://modelingbook.informatik.uni-ulm.de/

additionalmaterial#mutualexclusion

FinalDebitAmtLimitsCheck(pendgT ,Proceed)

let (otp, (debit , from, to, a)) = pendgT

let debitor = owner (from), creditor = owner (to)

if CanBeSpentBy(cc(debitor ), a) then

if CanBeCashedBy(cc(creditor ), a) then

if HasSellCapacityFor (a, cc(creditor )) then Proceed

else RejectTransDueTo(SellCapacityViolation, pendgT )

else RejectTransDueTo(UpperBalanceLimitErr , pendgT )

else RejectTransDueTo(AProblemAtDebitor , pendgT )

Send(ErrMsg(AvailBalanceViolation(debit , from, to, a)),

to debitor )

where RejectTransDueTo(reason, pendgT ) =

Send(ErrMsg(reason(debit , from, to, a)), to creditor )

status(pendgT ) := rejected

DebitAckRequestExpired component

If the debitor ’s AnswerMsg for the Debit request is not Received
within OtpLifetime, the Server will reject to perform the transaction
and inform creditor and debitor about it.

DebitAckRequestExpired =

if t ∈ PendingTransact and Expired(t) then

let t = (otp, (debit , from, to, amount)) -- NB: otp is unique

Delete(t ,PendingTransact)

Insert(t ,RejectedTransact)

status(t) := rejected

Send(RejectMsg(Debit , amount , debitor ), to creditor )

Send(OtpExpiredMsg(Debit , amount , creditor ), to debitor )

where debitor = owner (from), creditor = owner (to)

Two typical retail/consumer B2C operations

A consumer ∈ Consumer ∪ Consumer Verified purchase at a retail
in Retail ∪ Full paid in Euro triggers Sardex to HandleB2cEur:

the EuroAmount is recorded in the income(retail) account

a reward is issued as SRD-credit to a consumer ’s Sardex account

– A consumer ∈ Consumer remains anonymous (a card represents
its account) until it becomes a Consumer Verified member

• namely by a registration action that we do not model here

the Euro fee is paid by retail to the Sardex company

A consumer who is registered as Consumer Verified member can pay
also in SRD, thereby triggering a HandleB2cSrd operation.

NB. These 2 ops do not involve any group transfer or account type
checks so that we model them directly, not as Credit/Debit op instances.

HandleB2cEur component

HandleB2cEur = -- typically triggered via POS

if Received(B2CEurMsg(amount , consumer ), from retail)

and consumer ∈ Consumer ∪ Consumer Verified

and retail ∈ Retail ∪ Full then

if ThereIsEnoughPrepaidFeeFor (amount , retail)

and ThereAreEnoughCreditsFor (amount , retail)

then

HandleEuroPayment(amount , retail)

HandleRewardPayment(amount , retail , consumer )

HandleFeePayment(amount , retail)

else

IssueWarning(NotEnoughFundsInPrepaidOrCCAccounts)

Consume(Msg(amount , from consumer ))

HandleB2cEur macros

ThereIsEnoughPrepaidFeeFor (amount , retail) iff

balanceretail (prepaidretail ) ≥ B2CEuroFeeretail (amount)

ThereAreEnoughCreditsFor (amount , retail) iff

availableBalance(ccretail ) ≥ rewardRateretail (amount)

HandleEuroPayment(amount , retail) =

balance(incomeretail ) := balance(incomeretail ) + amount

HandleRewardPayment(amount , retail , cons) =

balance(ccretail ) := balance(ccretail )− rewardRateretail (amount)

balance(cccons) := balance(cccons) + rewardRateretail (amount)

HandleFeePayment(amount , retail) =

balance(prepaidretail ) :=

balance(prepaidretail )− B2CEuroFeeretail (amount)

HandleB2cSrd

HandleB2cSrd is triggered by a member of Consumer Verified at
a retail ∈ Retail ∪ Full when the retailer accepts a purchase the
member pays partly by accumulated SRD rewards, the rest in Euro.

HandleB2cSrd = -- amount is the total purchase value

if Received(B2CSrdMsg(amount , consumer ), from retail)

and consumer ∈ Consumer Verified

and retail ∈ Retail ∪ Full then

PayWithReward(amount , consumer , retail)

Consume(SrdB2CMsg(amount , consumer ))

where PayWithReward(amount , consumer , retail) =

balance(ccconsumer ) :=

balance(ccconsumer ))− acceptanceRate(amount)

balance(ccretail ) := balance(ccretail ) + acceptanceRate(amount)

MngrOps component RechargePrepaid

Upon the receipt of a fee prepayment by a retailer through normal
banking channels, the prepaid amount is added to prepaidretail .

– We skip the additional updates for double-entry bookkeeping.

Before prepaidretail reaches zero, a LowPrepayAlert msg is sent
to the retailer.

RechargePrepaid =

if Received(EuroFeePrepaymentMsg(amount , from retail))

and retail ∈ Retail ∪ Full

then

balance(prepaidretail ) := balance(prepaidretail ) + amount

Consume(EuroFeePrepaymentMsg(amount , from retail))

Components to handleAlerts

LowPrepayAlert =

forall retail ∈ Retail ∪ Full

if CloseToZero(balance(prepaidretail )) then

Send(PrepaymentAlertMsg(lowPrepaidBalance), to retail)

Analogously for other alerts concerning CC -accounts, e.g.

LowBalanceAlert iff availableBalance < lowBalanceAlert

-- small amount of money left for further purchases in SRD

HighBalanceAlert iff balance > 0 and

upperBalanceLimit − balance < highBalanceAlert

-- small space left for further sales in SRD

HighVolumeAlert iff availableSaleCapacity < highVolumeAlert

Recap HandleSardexOpns

HandleCreditPreviewReq

HandleCreditPerformReq

 --HandleCreditTransfer

HandleDebitPreviewReq

HandleDebitPerformReq

HandleDebitAckCompletion

 --HandleDebitTransfer

HandleB2cEur

HandleB2cSrd

RechargePrepaid

 --HandleB2CTransfer

HandleAlerts

The Sardex system of communicating agents

What makes Sardex (not the company, but the mutual credit system)
a system of communicating agents is the interaction of:

instances of MemberOpns each circuit member is equipped with

– those which trigger and support to HandleSardexOpns

program HandleSardexOpns the network server sardex executes

– in reality with many more ops than those shown here

operations of other agents we did not model here, e.g. of

– a software systemAdmin istrator

– brokers (which survey and support the well functioning of the circuit)

– agents belonging to associated circuits

Essentially, MemberOpns consists of rules which describe the Send
and Receive actions of circuit members, here concerning the
request/response msgs exchanged with HandleSardexOpns.

MemberOpns

Members supply their various request and response msgs by filling in
appropriate fields on a screen. The editing functionality is clear so that
we skip rules which describe how to prepare and send a
Credit/DebitPreviewReq(t) and similar requests.

For Credit/Debit Perform requests, the only relevant additional
constraint is that these requests are triggered by receiving an OK
message, sent by sardex , for the corresponding Preview request.

if Received(YouMayTriggerPerformReq(t), from sardex ) then

if transferKind(t) = credit then

Send(CreditPerformReq(t), to sardex )

if transferKind(t) = debit then

Send(DebitPerformReq(t), to sardex )

Consume(YouMayTriggerPerformReq(t))

MemberOpns (Cont’d)

In case of a Debit operation the debitor is expected to confirm a received
debit request by Sending a DebitAckMsg ; otherwise the debitor should
send a DebitRejectMsg to the network server.

if Received(AgreementReq(otp, transfer ), from sardex ) then

let (debit , from, to, amount) = transfer

if Agreed(amount , owner (to), otp)

then Send(DebitAckMsg(otp, transfer ), to sardex )

else Send(DebitRejectMsg(otp, transfer ), to sardex )

Consume(AgreementReq(otp, transfer ))

We skip the obvious rules for receiving an ErrMsg or pure information
msgs like RejectionInfo, OtpExpiredMsg , etc.

A characteristic refinement example: debt record tracking

The requirement: The balance of an account can go negative, but
the debt must be ‘paid back’ within 12 months of when it was incurred.

The debt is not towards Sardex S.p.A. and it is not bilateral towards a
single other member. Rather, the debt is towards the circuit as a whole.

Therefore, the indebted circuit member can ‘pay back’ its debt by
selling its products and services to other members using Sardex
accounts for the payment.

More precisely, every single debt incurred by a transaction is required

to be completely recovered, possibly in portions, within a
payBackWindow (currently of 12 months)

to remain recorded (for legal reasons) with its current rest debt value
(initially the debt incurred, eventually 0).

How to integrate the debt record tracking requirement

Algorithmic refinement idea.

For any new debt create a debtEntry , inserted into a DebtRecord
and consisting of: its birthTime, the debt triggering transfer
(opKind , from, to, amount) and the restDebt (initially debt).

When an account receives a positive credit amount transfer, call
UpdateDebtRecord to pay back the corresponding portion of
still not-completely-repayed debts in DebtRecord , proceeding
sequentially, starting from the oldest unpaid one.

Then the refinement (in fact a conservative refinement) is to simply

add CreateDebtEntry and UpdateDebtRecord

in parallel to the rules of the following components:

CompleteTransaction

HandleRewardPayment in HandleB2cEur

PayWithReward in HandleB2cSrd

CreateDebtEntry(transfer )

CreateDebtEntry(opKind , from, to, amount) =

if balance(from)− amount < 0 then

-- balance(from) turns negative

let debtEntry =new (DebtRecord)

birthTime(debtEntry) := now

transfer (debtEntry) := (opKind , from, to, amount)

restDebt(debtEntry) := | balance(from)− amount | if balance(from) > 0

amount else

NB. Because of the time parameter, two calls of
CreateDebtEntry(t) with the same t create different debt entries.

A coarser time scheme implies the need to keep the transfer
parameters distinct to identify the corresponding debt entries.

UpdateDebtRecord

UpdateDebtRecord(opKind , from, to, amount) =

if amount 6= 0 then -- balance(to) increases

let OpenDebtEntries = {e ∈ DebtRecord | restDebt(e) > 0}
Update(OpenDebtEntries , amount)

where Update(E , amount) = -- a recursive ASM

if E 6= ∅ then -- recursion termination condition

let entry = oldest(E ) -- element with earliest birthTime

if restDebt(entry) ≥ amount

then restDebt(entry) := restDebt(entry)− amount

else restDebt(entry) := 0

let restAmount = amount − restDebt(entry)

Update(E \ {entry}, restAmount) -- recursive call

For the definition of recursive ASMs see the References.

Extending Alerts by an UnpaidDebtAlert

The requirement that ‘the debt must be ‘paid back’ within 12 months
of when it was incurred’ triggers the question how to enforce the
‘must’.

The question motivates the introduction of an UnpaidDebtAlert for
each account involved. This can be done by simply adding to Alerts
a corresponding rule in parallel.

– NB. DebtRecord belongs to an account by which it is implicitly
parameterized. When we want to make this parameter explicit we
write DebtRecordacc or DebtRecord(acc).

UnpaidDebtAlert(acc) iff

forsome entry ∈ DebtRecordacc with restDebt(entry) > 0

now − birthtime(entry) IsCloseTo payBackWindow

Concluding remarks

The Sardex ASM describes a tiny subset of Sardex core services.

– But the model is effectively and easily extendable, due to its
component structure and its formulation by proceeding from abstract
to more and more detailed descriptions by stepwise refinement.

• In the INTERLACE project the stepwise refinement technique,
which we illustrated here for adding the debt record tracking, has
been used throughout to develop the model from scratch.

The model is a ground model, i.e. tailored to express the services of
the Sardex mutual credit system at the business logic level of
abstraction, avoiding any only code-related programming terms.

– The pseudocode character of ASMs permits to directly describe the
computational rules by which a process step changes the underlying
abstract state, without need to encode the involved state elements.

– The pseudocode form of ASMs supports the use of ground models as
guide for the implementation, as done in the INTERLACE project.

